Pensieve header: The full sl_2 invariant using the Drinfel'd double. Continues 2018-05/ybax.nb, Talks/StonyBrook-1805/ybax.nb, Projects/SL2Portfolio/Logoi.nb.

Profiling

In[∘]:=
```
Once[
  SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\SL2Invariant"];
  << KnotTheory`;
  << "../Profile/Profile.m";
 ];
BeginProfile[];
Once@PopupWindow[Button["Show Profile Monitor"],
  Dynamic[PrintProfile[], UpdateInterval → 3, TrackedSymbols → {}]]
```

```
Loading KnotTheory` version of January 20, 2015, 10:42:19.1122.
Read more at http://katlas.org/wiki/KnotTheory.
This is Profile.m of http://www.drorbn.net/AcademicPensieve/Projects/Profile/.
This version: June 2018. Original version: July 1994.
```

Out[∘]= Show Profile Monitor

External Utilities

In[∘]:=
```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Program

Program

Internal Utilities

Program

Canonical Form:

Program

In[∘]:=
```
CCF[ℰ_] := PP_CCF@ExpandDenominator@ExpandNumerator@PP_Together@Together[PP_Exp[
        Expand[ℰ] //. e^x_ e^y_ :> e^(x+y) /. e^x_ :> e^CCF[x]]];
CF[ℰ_List] := CF /@ ℰ;
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := PP_CF@Module[
    {vs = Cases[ℰ, (y | b | t | a | x | η | β | τ | α | ξ)_, ∞] ⋃ {y, b, t, a, x, η, β, τ, α, ξ}},
    Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) :> CCF[c] (Times @@ vs^ps)]
   ];
```

Program

The Kronecker δ:

Program

In[∘]:=
```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Program

Equality, multiplication, and degree-adjustment of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

Program

In[∘]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
𝔼[L_, Q_, P_]_{$k_} := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Program

Zip and Bind

Program

Variables and their duals:

Program

In[∘]:=
```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_{i_})* := (u*)_i;
```

Program

Finite Zips:

Program

In[∘]:=
```
collect[sd_SeriesData, ζ_] := MapAt[collect[#, ζ] &, sd, 3];
collect[ℰ_, ζ_] := PP_Collect@Collect[ℰ, ζ];
Zip_{}[P_] := P; Zip_{ζ_,ζs___}[P_] := PP_Zip[
  (collect[P // Zip_{ζs}, ζ] /. f_. ζ^d_. :> ∂_{ζ*,d} f) /. ζ* → 0]
```

Program

QZip implements the “Q-level zips” on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the L variables as scalars.

Program

In[∘]:=
```
QZip_{ζs_List}@𝔼[L_, Q_, P_] := PP_QZip@Module[{ζ, z, zs, c, ys, ηs, qt, zrule, ζrule},
    zs = Table[ζ*, {ζ, ζs}];
    c = CF[Q /. Alternatives @@ (ζs ⋃ zs) → 0];
    ys = CF@Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = CF@Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
    qt = CF@Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → CF[qt.(zs + ys)]];
    ζrule = Thread[ζs → ζs + ηs.qt];
    CF /@ 𝔼[L, c + ηs.qt.ys, Det[qt] Zip_{ζs}[P /. (zrule ⋃ ζrule)]] ];
```

Program

Upper to lower and lower to Upper:

Program

In[∘]:=
```
U2l = {B_{i_}^{p_.} → e^{-p ħ γ b_i}, B^{p_.} → e^{-p ħ γ b}, T_{i_}^{p_.} → e^{p ħ t_i}, T^{p_.} → e^{p ħ t}, 𝒜_{i_}^{p_.} → e^{p γ α_i}, 𝒜^{p_.} → e^{p γ α}};
l2U = {e^{c_. b_{i_}+d_.} :> B_i^{-c/(ħ γ)} e^d, e^{c_. b+d_.} :> B^{-c/(ħ γ)} e^d,
   e^{c_. t_{i_}+d_.} :> T_i^{c/ħ} e^d, e^{c_. t+d_.} :> T^{c/ħ} e^d,
   e^{c_. α_{i_}+d_.} :> 𝒜_i^{c/γ} e^d, e^{c_. α+d_.} :> 𝒜^{c/γ} e^d,
   e^{ℰ_} :> e^{Expand@ℰ}};
```

Program

LZip implements the “L-level zips” on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard all of Pe^Q as a single”P”. Here the z’s are b and α and the ζ’s are β and a.

Program

In[∘]:=
```
LZip_{ζs_List}@𝔼[L_, Q_, P_] :=
  PP_LZip@Module[{ζ, z, zs, c, ys, ηs, lt, zrule, Zrule, ζrule, Q1, EEQ, EQ},
    zs = Table[ζ*, {ζ, ζs}];
    c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
    ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
    lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} L, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → lt.(zs + ys)];
    Zrule = zrule /. r_Rule :>
       ((U = r⟦1⟧ /. {b → B, t → T, α → 𝒜}) → (U /. U2l /. r //. l2U)); (* not used *)
    ζrule = Thread[ζs → ζs + ηs.lt];
    Q1 = Q /. U2l /. (zrule ⋃ ζrule);
    EEQ[ps___] := EEQ[ps] = PP_"EEQ"@(CF[e^{-Q1} D[e^{Q1}, Sequence @@ Thread[{zs, {ps}}]]] /.
          Alternatives @@ zs → 0 //. l2U);
    CF /@ ((*CF/@*)𝔼[
        c + ηs.lt.ys, Q1 /. Alternatives @@ zs → 0,
        Det[lt] (Zip_{ζs}[(EQ @@ zs) (P /. U2l /. (zrule ⋃ ζrule))] /.
           Derivative[ps___][EQ][___] :> EEQ[ps] /. _EQ → 1)
      ] //. l2U)
   ];
```

Program

In[∘]:=
```
B_{}[L_, R_] := L R;
B_{is__}[L_𝔼, R_𝔼] := PP_B@Module[{n},
    Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
     ] // LZip_{Join@@Table[{β_{n@i},τ_{n@i},a_{n@i}},{i,{is}}]} // QZip_{Join@@Table[{ξ_{n@i},y_{n@i}},{i,{is}}]} ];
B_{is___}[L_, R_] := B_{{is}}[L, R];
```

Program

𝔼 morphisms with domain and range.

Program

In[∘]:=
```
B_{is_List}[𝔼_{d1_→r1_}[L1_, Q1_, P1_], 𝔼_{d2_→r2_}[L2_, Q2_, P2_]] :=
  𝔼_{(d1⋃Complement[d2,is])→(r2⋃Complement[r1,is])} @@ B_{is}[𝔼[L1, Q1, P1], 𝔼[L2, Q2, P2]];
𝔼_{d1_→r1_}[L1_, Q1_, P1_] // 𝔼_{d2_→r2_}[L2_, Q2_, P2_] :=
  B_{r1⋂d2}[𝔼_{d1→r1}[L1, Q1, P1], 𝔼_{d2→r2}[L2, Q2, P2]];
𝔼_{d1_→r1_}[L1_, Q1_, P1_] ≡ 𝔼_{d2_→r2_}[L2_, Q2_, P2_] ^:=
  (d1 == d2) ∧ (r1 == r2) ∧ (𝔼[L1, Q1, P1] ≡ 𝔼[L2, Q2, P2]);
𝔼_{d1_→r1_}[L1_, Q1_, P1_] 𝔼_{d2_→r2_}[L2_, Q2_, P2_] ^:=
  𝔼_{(d1⋃d2)→(r1⋃r2)} @@ (𝔼[L1, Q1, P1] 𝔼[L2, Q2, P2]);
𝔼_{d_→r_}[L_, Q_, P_]_{$k_} := 𝔼_{d→r} @@ 𝔼[L, Q, P]_{$k};
𝔼_[ℰ___][i_] := {ℰ}〚i〛;
```

Program

“Define” code

Program

Define[lhs = rhs, ...] defines the lhs to be rhs, except that rhs is computed only once for each value of $k. Fancy Mathematica not for the faint of heart. Most readers should ignore.

Program

In[∘]:=
```
SetAttributes[Define, HoldAll];
Define[def_, defs__] := (Define[def]; Define[defs];);
Define[op_{is__} = ℰ_] := Module[{SD, ii, jj, kk, isp, nis, nisp, sis}, Block[{i, j, k},
    ReleaseHold[Hold[
        SD[op_{nisp,$k_Integer}, PP_{Boot}@Block[{i, j, k}, op_{isp,$k} = ℰ; op_{nis,$k}]];
        SD[op_{isp}, op_{{is},$k}]; SD[op_{sis__}, op_{{sis}}];
      ] /. {SD → SetDelayed,
        isp → {is} /. {i → i_, j → j_, k → k_},
        nis → {is} /. {i → ii, j → jj, k → kk},
        nisp → {is} /. {i → ii_, j → jj_, k → kk_}
      }] ]]
```

Program

Booting Up

Program

In[∘]:=
```
$k = 2; (*ħ=γ=1;*)
```

Program

In[∘]:=
```
Define[am_{i,j→k} = 𝔼_{{i,j}→{k}}[(α_i + α_j) a_k, (e^{-γ α_j} ξ_i + ξ_j) x_k, 1]_{$k},
  bm_{i,j→k} = 𝔼_{{i,j}→{k}}[(β_i + β_j) b_k, (η_i + η_j) y_k, e^{(e^{-ϵ β_i}-1) η_j y_k}]_{$k}]
```

Program

In[∘]:=
```
Define[R_{i,j} = 𝔼_{{}→{i,j}}[ħ a_j b_i, ħ x_j y_i, ⅇ^(Sum_{k=2}^{$k+1} (1 - ⅇ^{γ ϵ ħ})^k (ħ y_i x_j)^k / (k (1 - ⅇ^{k γ ϵ ħ})))]_{$k},
  R̄_{i,j} = 𝔼_{{}→{i,j}}[-ħ a_j b_i, -ħ x_j y_i / B_i, 1 + If[$k == 0, 0, (R̄_{{i,j},$k-1})_{$k}[3] -
      (((R̄_{{i,j},0})_{$k} R_{1,2} (R̄_{{3,4},$k-1})_{$k}) // (bm_{i,1→i} am_{j,2→j}) // (bm_{i,3→i} am_{j,4→j}))[3]]],
  P_{i,j} = 𝔼_{{i,j}→{}}[β_i α_j / ħ, η_i ξ_j / ħ, 1 + If[$k == 0, 0, (P_{{i,j},$k-1})_{$k}[3] -
      (R_{1,2} // ((P_{{1,j},0})_{$k} (P_{{i,2},$k-1})_{$k}))[3]]]]
```

Program

In[∘]:=
```
Define[aS_j = R̄_{i,j} ~ B_i ~ P_{i,j},
  aS̄_i = 𝔼_{{i}→{i}}[-a_i α_i, -x_i 𝒜_i ξ_i, 1 + If[$k == 0, 0, (aS̄_{{i},$k-1})_{$k}[3] -
      ((aS̄_{{i},0})_{$k} ~ B_i ~ aS_i ~ B_i ~ (aS̄_{{i},$k-1})_{$k})[3]]]]
```

Program

In[∘]:=
```
Define[bS_i = R_{i,1} ~ B_1 ~ aS_1 ~ B_1 ~ P_{i,1},
  bS̄_i = R_{i,1} ~ B_1 ~ aS̄_1 ~ B_1 ~ P_{i,1},
  aΔ_{i→j,k} = (R_{1,j} R_{2,k}) // bm_{1,2→3} // P_{3,i},
  bΔ_{i→j,k} = (R_{j,1} R_{k,2}) // am_{1,2→3} // P_{i,3}]
```

Program

In[∘]:=
```
Define[dm_{i,j→k} = (𝔼_{{i,j}→{i,j}}[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1]
      (aΔ_{i→1,2} // aΔ_{2→2,3} // aS̄_3) (bΔ_{j→-1,-2} // bΔ_{-2→-2,-3})) // (P_{-1,3} P_{-3,1} am_{2,j→k} bm_{i,-2→k}),
  dS_i = 𝔼_{{i}→{1,2}}[β_i b_1 + α_i a_2, η_i y_1 + ξ_i x_2, 1] // (bS̄_1 aS_2) // dm_{2,1→i},
  dΔ_{i→j,k} = (bΔ_{i→3,1} aΔ_{i→2,4}) // (dm_{3,4→k} dm_{1,2→j})]
```

Program

In[∘]:=
```
Define[C_i = 𝔼_{{}→{i}}[0, 0, B_i^{1/2} ⅇ^{-ħ ϵ a_i/2}]_{$k},
  C̄_i = 𝔼_{{}→{i}}[0, 0, B_i^{-1/2} ⅇ^{ħ ϵ a_i/2}]_{$k},
  Kink_i = (R_{1,3} C̄_2) // dm_{1,2→1} // dm_{1,3→i},
  K̄ink_i = (R̄_{1,3} C_2) // dm_{1,2→1} // dm_{1,3→i}]
```

Program

Note. $t == \epsilon a - \gamma b$ and $b == -t/\gamma + \epsilon a/\gamma$.

Program

In[∘]:=
```
Define[b2t_i = 𝔼_{{i}→{i}}[α_i a_i - β_i t_i / γ, ξ_i x_i + η_i y_i, ⅇ^{ϵ β_i a_i/γ}]_{$k},
  t2b_i = 𝔼_{{i}→{i}}[α_i a_i - τ_i γ b_i, ξ_i x_i + η_i y_i, ⅇ^{ϵ τ_i a_i}]_{$k}]
```

Testing

In[∘]:=
```
Block[{$k = 1}, {
   am → am_{i,j→k}, bm → bm_{i,j→k}, dm → dm_{i,j→k}, R → R_{i,j}, R̄ → R̄_{i,j}, P → P_{i,j},
   aS → aS_i, aS̄ → aS̄_i, bS → bS_i, bS̄ → bS̄_i, dS → dS_i, aΔ → aΔ_{i→j,k}, bΔ → bΔ_{i→j,k},
   dΔ → dΔ_{i→j,k}, C → C_i, C̄ → C̄_i, Kink → Kink_i, K̄ink → K̄ink_i, b2t → b2t_i, t2b → t2b_i
  }] //
 Column
```

Out[∘]=

$$am \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[a_k\,(\alpha_i+\alpha_j),\ x_k\left(e^{-\gamma\alpha_j}\,\xi_i+\xi_j\right),\ 1\right]$$

$$bm \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[b_k\,(\beta_i+\beta_j),\ y_k\,(\eta_i+\eta_j),\ 1-y_k\,\beta_i\,\eta_j\,\epsilon+O[\epsilon]^2\right]$$

$$dm \to \mathbb{E}_{\{i,j\}\to\{k\}}\Big[a_k\,\alpha_i+a_k\,\alpha_j+b_k\,\beta_i+b_k\,\beta_j,\ y_k\,\eta_i+\frac{y_k\,\eta_j}{\mathcal{A}_i}+\frac{x_k\,\xi_i}{\mathcal{A}_j}+\frac{(1-B_k)\,\eta_j\,\xi_i}{\hbar}+x_k\,\xi_j,$$
$$1+\left(-\frac{y_k\,\beta_i\,\eta_j}{\mathcal{A}_i}-\frac{x_k\,\beta_j\,\xi_i}{\mathcal{A}_j}+a_k\,B_k\,\eta_j\,\xi_i+\frac{\gamma\,\hbar\,x_k\,y_k\,\eta_j\,\xi_i}{\mathcal{A}_i\,\mathcal{A}_j}+\frac{(\gamma-3\,\gamma\,B_k)\,y_k\,\eta_j^2\,\xi_i}{2\,\mathcal{A}_i}+\frac{(\gamma-3\,\gamma\,B_k)\,x_k\,\eta_j\,\xi_i^2}{2\,\mathcal{A}_j}+\frac{\left(\gamma-4\,\gamma\,B_k+3\,\gamma\,B_k^2\right)\eta_j^2\,\xi_i^2}{4\,\hbar}\right)\epsilon+$$
$$O[\epsilon]^2\Big]$$

$$R \to \mathbb{E}_{\{\}\to\{i,j\}}\left[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ 1-\frac{1}{4}\left(\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{R} \to \mathbb{E}_{\{\}\to\{i,j\}}\left[-\hbar\,a_j\,b_i,\ -\frac{\hbar\,x_j\,y_i}{B_i},\ 1+\left(-\frac{\hbar^2\,a_j\,x_j\,y_i}{B_i}-\frac{3\,\gamma\,\hbar^3\,x_j^2\,y_i^2}{4\,B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$P \to \mathbb{E}_{\{i,j\}\to\{\}}\left[\frac{\alpha_j\,\beta_i}{\hbar},\ \frac{\eta_i\,\xi_j}{\hbar},\ 1+\frac{\gamma\,\eta_i^2\,\xi_j^2\,\epsilon}{4\,\hbar}+O[\epsilon]^2\right]$$

$$aS \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\left(-\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\frac{1}{2}\,\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{aS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\left(\gamma\,\hbar\,x_i\,\mathcal{A}_i\,\xi_i-\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\frac{1}{2}\,\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$bS \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\left(-\frac{y_i\,\beta_i\,\eta_i}{B_i}-\frac{\gamma\,\hbar\,y_i^2\,\eta_i^2}{2\,B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{bS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\left(\frac{\gamma\,\hbar\,y_i\,\eta_i}{B_i}-\frac{y_i\,\beta_i\,\eta_i}{B_i}-\frac{\gamma\,\hbar\,y_i^2\,\eta_i^2}{2\,B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$dS \to \mathbb{E}_{\{i\}\to\{i\}}\Big[-a_i\,\alpha_i-b_i\,\beta_i,\ -\frac{y_i\,\mathcal{A}_i\,\eta_i}{B_i}-x_i\,\mathcal{A}_i\,\xi_i+\frac{(\mathcal{A}_i-B_i\,\mathcal{A}_i)\,\eta_i\,\xi_i}{\hbar\,B_i},$$
$$1+\Big(\frac{\gamma\,\hbar\,y_i\,\mathcal{A}_i\,\eta_i}{B_i}-\frac{y_i\,\mathcal{A}_i\,\beta_i\,\eta_i}{B_i}-\frac{\gamma\,\hbar\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2}{2\,B_i^2}-\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-x_i\,\mathcal{A}_i\,\beta_i\,\xi_i+\frac{a_i\,\mathcal{A}_i\,\eta_i\,\xi_i}{B_i}-$$
$$\frac{\gamma\,\hbar\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i}{B_i}+\frac{(-\gamma\,\mathcal{A}_i+\gamma\,B_i\,\mathcal{A}_i)\,\eta_i\,\xi_i}{B_i}+\frac{(\mathcal{A}_i-B_i\,\mathcal{A}_i)\,\beta_i\,\eta_i\,\xi_i}{\hbar\,B_i}+\frac{y_i\left(3\,\gamma\,\mathcal{A}_i^2-\gamma\,B_i\,\mathcal{A}_i^2\right)\eta_i^2\,\xi_i}{2\,B_i^2}-$$
$$\frac{1}{2}\,\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2+\frac{x_i\left(3\,\gamma\,\mathcal{A}_i^2-\gamma\,B_i\,\mathcal{A}_i^2\right)\eta_i\,\xi_i^2}{2\,B_i}+\frac{\left(-3\,\gamma\,\mathcal{A}_i^2+4\,\gamma\,B_i\,\mathcal{A}_i^2-\gamma\,B_i^2\,\mathcal{A}_i^2\right)\eta_i^2\,\xi_i^2}{4\,\hbar\,B_i^2}\Big)\epsilon+O[\epsilon]^2\Big]$$

$$a\Delta \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[a_j\,\alpha_i+a_k\,\alpha_i,\ x_j\,\xi_i+x_k\,\xi_i,\ 1+\left(-\hbar\,a_j\,x_k\,\xi_i+\frac{1}{2}\,\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$b\Delta \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[b_j\,\beta_i+b_k\,\beta_i,\ B_k\,y_j\,\eta_i+y_k\,\eta_i,\ 1+\frac{1}{2}\,\gamma\,\hbar\,B_k\,y_j\,y_k\,\eta_i^2\,\epsilon+O[\epsilon]^2\right]$$

$$d\Delta \to \mathbb{E}_{\{i\}\to\{j,k\}}\Big[a_j\,\alpha_i+a_k\,\alpha_i+b_j\,\beta_i+b_k\,\beta_i,\ y_j\,\eta_i+B_j\,y_k\,\eta_i+x_j\,\xi_i+x_k\,\xi_i,$$
$$1+\left(\frac{1}{2}\,\gamma\,\hbar\,B_j\,y_j\,y_k\,\eta_i^2-\hbar\,a_j\,x_k\,\xi_i+\frac{1}{2}\,\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\Big]$$

$$C \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \sqrt{B_i}-\frac{1}{2}\left(\hbar\,a_i\,\sqrt{B_i}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{C} \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}}+\frac{\hbar\,a_i\,\epsilon}{2\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$Kink \to \mathbb{E}_{\{\}\to\{i\}}\left[\hbar\,a_i\,b_i,\ \hbar\,x_i\,y_i,\ \frac{1}{\sqrt{B_i}}+\left(\frac{\hbar\,a_i}{2\,\sqrt{B_i}}-\frac{\gamma\,\hbar^3\,x_i^2\,y_i^2}{4\,\sqrt{B_i}}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{Kink} \to \mathbb{E}_{\{\}\to\{i\}}\left[-\hbar\,a_i\,b_i,\ -\frac{\hbar\,x_i\,y_i}{B_i},\ \sqrt{B_i}+\left(-\frac{1}{2}\,\hbar\,a_i\,\sqrt{B_i}-\frac{\hbar^2\,a_i\,x_i\,y_i}{\sqrt{B_i}}-\frac{3\,\gamma\,\hbar^3\,x_i^2\,y_i^2}{4\,B_i^{3/2}}\right)\epsilon+O[\epsilon]^2\right]$$

$$b2t \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\,\alpha_i-\frac{t_i\,\beta_i}{\gamma},\ y_i\,\eta_i+x_i\,\xi_i,\ 1+\frac{a_i\,\beta_i\,\epsilon}{\gamma}+O[\epsilon]^2\right]$$

$$t2b \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\,\alpha_i-\gamma\,b_i\,\tau_i,\ y_i\,\eta_i+x_i\,\xi_i,\ 1+a_i\,\tau_i\,\epsilon+O[\epsilon]^2\right]$$

Check that on the generators this agrees with our conventions in the handout:

In[∘]:=
```
Timing@
 {{"[a,x]" → ((𝔼_{}→{1,2}[0, 0, a2 x1] // am1,2→1)[3] - (𝔼_{}→{1,2}[0, 0, a1 x2] // am1,2→1)[3]),
    "[b,y]" → ((𝔼_{}→{1,2}[0, 0, y2 b1] // bm1,2→1)[3] - (𝔼_{}→{1,2}[0, 0, y1 b2] // bm1,2→1)[3])} /.
   z_1 → z,
  {"Δ[y]" → Last[𝔼_{}→{1}[0, 0, y1] ~B1~ bΔ1→1,2],
   "Δ[b]" → Last[𝔼_{}→{1}[0, 0, b1] ~B1~ bΔ1→1,2],
   "Δ[a]" → Last[𝔼_{}→{1}[0, 0, a1] ~B1~ aΔ1→1,2],
   "Δ[x]" → Last[𝔼_{}→{1}[0, 0, x1] ~B1~ aΔ1→1,2]},
  {
    "S(a)" → ((𝔼_{}→{1}[0, 0, a1] ~B1~ aS1)[3]),
    "S(x)" → ((𝔼_{}→{1}[0, 0, x1] ~B1~ aS1)[3]),
    "S(b)" → ((𝔼_{}→{1}[0, 0, b1] ~B1~ bS1)[3]),
    "S(y)" → ((𝔼_{}→{1}[0, 0, y1] ~B1~ bS1)[3])
   } /. z_1 → z}
```

Out[∘]= $\{1., \{\{[a,x] \to -x\,\gamma,\ [b,y] \to -y\,\epsilon + O[\epsilon]^3\},\ \{\Delta[y] \to (B_2\,y_1 + y_2) + O[\epsilon]^3,\ \Delta[b] \to (b_1 + b_2) + O[\epsilon]^3,$
$\Delta[a] \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta[x] \to (x_1 + x_2) - \hbar\,a_1\,x_2\,\epsilon + \frac{1}{2}\hbar^2\,a_1^2\,x_2\,\epsilon^2 + O[\epsilon]^3\},\ \{S(a) \to -a + O[\epsilon]^3,$
$S(x) \to -x - a\,x\,\hbar\,\epsilon - \frac{1}{2}(a^2\,x\,\hbar^2)\,\epsilon^2 + O[\epsilon]^3,\ S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + O[\epsilon]^3\}\}\}$

Hopf algebra axioms on both sides separately.

Associativity of am and bm:

In[∘]:=
```
Timing@Block[{$k = 3},
  HL /@ {(am1,2→1 // am1,3→1) ≡ (am2,3→2 // am1,2→1), (bm1,2→1 // bm1,3→1) ≡ (bm2,3→2 // bm1,2→1)}
 ]
```

Out[∘]= {0.171875, {True, True}}

R and *P* are inverses:

In[∘]:=
```
Timing@Block[{$k = 3}, {Ri,j, Pi,k, HL[(Ri,j // Pi,k) ≡ 𝔼_{k}→{j}[aj αk, xj ξk, 1]]}]
```

Out[∘]= $\{0.125, \{\mathbb{E}_{\{\}\to\{i,j\}}[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ 1 - \frac{1}{4}(\gamma\,\hbar^3\,x_j^2\,y_i^2)\,\epsilon + \left(\frac{1}{9}\gamma^2\,\hbar^5\,x_j^3\,y_i^3 + \frac{1}{32}\gamma^2\,\hbar^6\,x_j^4\,y_i^4\right)\epsilon^2 +$
$\frac{1}{1152}(24\,\gamma^3\,\hbar^5\,x_j^2\,y_i^2 - 72\,\gamma^3\,\hbar^7\,x_j^4\,y_i^4 - 32\,\gamma^3\,\hbar^8\,x_j^5\,y_i^5 - 3\,\gamma^3\,\hbar^9\,x_j^6\,y_i^6)\,\epsilon^3 + O[\epsilon]^4],$
$\mathbb{E}_{\{i,k\}\to\{\}}[\frac{\alpha_k\,\beta_i}{\hbar},\ \frac{\eta_i\,\xi_k}{\hbar},\ 1 + \frac{\gamma\,\eta_i^2\,\xi_k^2\,\epsilon}{4\,\hbar} + \frac{1}{288\,\hbar^2}(36\,\gamma^2\,\hbar^2\,\eta_i^2\,\xi_k^2 + 40\,\gamma^2\,\hbar\,\eta_i^3\,\xi_k^3 + 9\,\gamma^2\,\eta_i^4\,\xi_k^4)\,\epsilon^2 +$
$\left(\frac{1}{24}\gamma^3\,\hbar\,\eta_i^2\,\xi_k^2 + \frac{1}{6}\gamma^3\,\eta_i^3\,\xi_k^3 + \frac{13\,\gamma^3\,\eta_i^4\,\xi_k^4}{96\,\hbar} + \frac{5\,\gamma^3\,\eta_i^5\,\xi_k^5}{144\,\hbar^2} + \frac{\gamma^3\,\eta_i^6\,\xi_k^6}{384\,\hbar^3}\right)\epsilon^3 + O[\epsilon]^4],\ \text{True}\}\}$

as and $\overline{\text{aS}}$ are inverses, bs and $\overline{\text{bS}}$ are inverses:

In[∘]:=
```
Timing[HL /@ {(aS̄1 // aS1) ≡ 𝔼_{1}→{1}[a1 α1, x1 ξ1, 1], (bS̄1 // bS1) ≡ 𝔼_{1}→{1}[b1 β1, y1 η1, 1]}]
```

Out[∘]= {0.359375, {True, True}}

(co)-associativity on both sides

```
In[ ]:= Timing[
  HL /@ {(aΔ_{1→1,2} // aΔ_{2→2,3}) ≡ (aΔ_{1→1,3} // aΔ_{1→1,2}), (bΔ_{1→1,2} // bΔ_{2→2,3}) ≡ (bΔ_{1→1,3} // bΔ_{1→1,2}),
    (am_{1,2→1} // am_{1,3→1}) ≡ (am_{2,3→2} // am_{1,2→1}), (bm_{1,2→1} // bm_{1,3→1}) ≡ (bm_{2,3→2} // bm_{1,2→1})}]
Out[ ]= {0.390625, {True, True, True, True}}
```

Δ is an algebra morphism

```
In[ ]:= Timing[HL /@ {(am_{1,2→1} // aΔ_{1→1,2}) ≡ ((aΔ_{1→1,3} aΔ_{2→2,4}) // (am_{3,4→2} am_{1,2→1})),
    (bm_{1,2→1} // bΔ_{1→1,2}) ≡ ((bΔ_{1→1,3} bΔ_{2→2,4}) // (bm_{3,4→2} bm_{1,2→1}))}]
Out[ ]= {0.625, {True, True}}
```

An explicit formula for aS_i

```
In[ ]:= Timing@Block[{$k = 4}, HL[aS_i ≡ (𝔼_{{i}→{i,j}}[-α_i a_j, -ξ_i x_i,
      Sum[Expand[(ⅇ^{ξ_i x_i} (-ℏ γ ϵ)^k)/(2^k k!) Nest[Expand[x_i^2 ∂_{{x_i,2}} #] &, ⅇ^{-ξ_i ⅇ^{ℏ ϵ a_i} x_i}, k]], {k, 0, $k}]]_{$k} //
    am_{i,j→i})]]
Out[ ]= {3.60938, True}
```

S is convolution inverse of id

```
In[ ]:= Timing[HL[# ≡ 𝔼_{{1}→{1}}[0, 0, 1]] & /@ {
    (aΔ_{1→1,2} ~ B_1 ~ aS_1) ~ B_{1,2} ~ am_{1,2→1}, (aΔ_{1→1,2} ~ B_2 ~ aS_2) ~ B_{1,2} ~ am_{1,2→1},
    (bΔ_{1→1,2} ~ B_1 ~ bS_1) ~ B_{1,2} ~ bm_{1,2→1}, (bΔ_{1→1,2} ~ B_2 ~ bS_2) ~ B_{1,2} ~ bm_{1,2→1}}]
Out[ ]= {0.484375, {True, True, True, True}}
```

But not with the opposite product:

```
In[ ]:= Timing[Short[# ≡ 𝔼_{{1}→{1}}[0, 0, 1]] & /@ {
    (aΔ_{1→1,2} ~ B_1 ~ aS_1) ~ B_{1,2} ~ am_{2,1→1}, (aΔ_{1→1,2} ~ B_2 ~ aS_2) ~ B_{1,2} ~ am_{2,1→1},
    (bΔ_{1→1,2} ~ B_1 ~ bS_1) ~ B_{1,2} ~ bm_{2,1→1}, (bΔ_{1→1,2} ~ B_2 ~ bS_2) ~ B_{1,2} ~ bm_{2,1→1}}]
```

Out[ ]= $\{0.640625, \{\frac{1}{2}\left(-2\,\gamma\,\epsilon\,\hbar\,x_1\,\mathcal{A}_1\,\xi_1+\gamma^2\,\epsilon^2\,\hbar^2\,x_1\,\mathcal{A}_1\,\xi_1-\ll 1\gg+2\,\gamma^2\,\epsilon^2\,\hbar^2\,x_1^2\,\mathcal{A}_1^2\,\xi_1^2\right)==0,$

$\frac{1}{2}\left(-2\,\gamma\,\epsilon\,\hbar\,x_1\,\xi_1-\gamma^2\,\epsilon^2\,\hbar^2\,x_1\,\xi_1+2\,\gamma^2\,\epsilon^2\,\hbar^2\,x_1^2\,\xi_1^2\right)==0,$

$\frac{1}{2}\left(-2\,\gamma\,\epsilon\,\hbar\,y_1\,\eta_1-\gamma^2\,\epsilon^2\,\hbar^2\,y_1\,\eta_1+2\,\gamma^2\,\epsilon^2\,\hbar^2\,y_1^2\,\eta_1^2\right)==0,$

$\frac{-2\,\gamma\,\epsilon\,\hbar\,B_1\,y_1\,\eta_1+\ll 3\gg+2\,\gamma^2\ll 3\gg\,\eta_1^2}{2\,B_1^2}==0\}\}$

S is an algebra anti-(co)morphism

```
In[ ]:= Timing[HL /@ {am_{1,2→1} ~ B_1 ~ aS_1 ≡ (aS_1 aS_2) ~ B_{1,2} ~ am_{2,1→1}, bm_{1,2→1} ~ B_1 ~ bS_1 ≡ (bS_1 bS_2) ~ B_{1,2} ~ bm_{2,1→1},
    aS_1 ~ B_1 ~ aΔ_{1→1,2} ≡ aΔ_{1→2,1} ~ B_{1,2} ~ (aS_1 aS_2), bS_1 ~ B_1 ~ bΔ_{1→1,2} ≡ bΔ_{1→2,1} ~ B_{1,2} ~ (bS_1 bS_2)}]
Out[ ]= {0.734375, {True, True, True, True}}
```

Pairing axioms

```
In[∘]:= Timing[HL /@ {(bm1,2→1 𝔼{3}→{3}[α3 a3, ξ3 x3, 1]) ~B1,3~P1,3 ≡
          (𝔼{1}→{1}[β1 b1, η1 y1, 1] 𝔼{2}→{2}[β2 b2, η2 y2, 1] aΔ3→4,5) ~B1,4~P1,4~B2,5~P2,5,
        (bΔ1→1,2 𝔼{3}→{3}[α3 a3, ξ3 x3, 1] 𝔼{4}→{4}[α4 a4, ξ4 x4, 1]) ~B1,3~P1,3~B2,4~P2,4 ≡
          (𝔼{1}→{1}[β1 b1, η1 y1, 1] am3,4→3) ~B1,3~P1,3 }]
Out[∘]= {0.34375, {True, True}}
```

```
In[∘]:= Timing[HL /@ {((bS1 𝔼{2}→{2}[α2 a2, ξ2 x2, 1]) // P1,2) ≡ ((𝔼{1}→{1}[β1 b1, η1 y1, 1] aS2) // P1,2),
        (b̄S̄1 𝔼{2}→{2}[α2 a2, ξ2 x2, 1]) ~B1,2~P1,2 ≡ (𝔼{1}→{1}[β1 b1, η1 y1, 1] āS̄2) ~B1,2~P1,2}]
Out[∘]= {0.28125, {True, True}}
```

Tests for the double.

Check the double formulas on the generators agree with SL2Portfolio.pdf:

```
In[∘]:= Timing@{{
          "[a,y]" →
           ((𝔼{}→{1,2}[0, 0, y2 a1] ~B1,2~dm1,2→1)[3] - (𝔼{}→{1,2}[0, 0, y1 a2] ~B1,2~dm1,2→1)[3]),
          "[b,x]"  → ((𝔼{}→{1,2}[0, 0, x2 b1] ~B1,2~dm1,2→1)[3] -
             (𝔼{}→{1,2}[0, 0, x1 b2] ~B1,2~dm1,2→1)[3]),
          "xy-qyx" → ((𝔼{}→{1,2}[0, 0, x1 y2] ~B1,2~dm1,2→1)[3] -
             (1 + ϵ) (𝔼{}→{1,2}[0, 0, y1 x2] ~B1,2~dm1,2→1)[3])
         } /. {z_1 → z} // Expand // Factor,
      {
        "Δ(a)" → ((𝔼{}→{1}[0, 0, a1] ~B1~dΔ1→1,2)[3]),
        "Δ(x)" → ((𝔼{}→{1}[0, 0, x1] ~B1~dΔ1→1,2)[3]),
        "Δ(b)" → ((𝔼{}→{1}[0, 0, b1] ~B1~dΔ1→1,2)[3]),
        "Δ(y)" → ((𝔼{}→{1}[0, 0, y1] ~B1~dΔ1→1,2)[3])
       } // Simplify,
      {
         "S(a)" → ((𝔼{}→{1}[0, 0, a1] ~B1~dS1)[3]),
         "S(x)" → ((𝔼{}→{1}[0, 0, x1] ~B1~dS1)[3]),
         "S(b)" → ((𝔼{}→{1}[0, 0, b1] ~B1~dS1)[3]),
         "S(y)" → ((𝔼{}→{1}[0, 0, y1] ~B1~dS1)[3])
       } /. {z_1 → z} // Simplify
     }
```

Out[∘]= $\{4.14063, \{\{[a,y] \to -y\gamma + O[\epsilon]^3,\ [b,x] \to x\epsilon + O[\epsilon]^3,$

$\text{xy-qyx} \to \frac{1-B}{\hbar} + (a B - x y + x y \gamma \hbar)\,\epsilon + \left(-\frac{1}{2} a^2 B \hbar + \frac{1}{2} x y \gamma^2 \hbar^2\right)\epsilon^2 + O[\epsilon]^3\},$

$\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta(x) \to (x_1 + x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3,$

$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta(y) \to (y_1 + B_1 y_2) + O[\epsilon]^3\},$

$\{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - a x \hbar \epsilon - \frac{1}{2}(a^2 x \hbar^2)\,\epsilon^2 + O[\epsilon]^3,$

$S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + \frac{y\gamma\hbar\epsilon}{B} - \frac{(y\gamma^2\hbar^2)\,\epsilon^2}{2B} + O[\epsilon]^3\}\}\}$

(co)-associativity

In[∘]:=
```
Timing[
  HL /@ {(dΔ1→1,2 // dΔ2→2,3) ≡ (dΔ1→1,3 // dΔ1→1,2), (dm1,2→1 // dm1,3→1) ≡ (dm2,3→2 // dm1,2->1)}]
```
Out[∘]= {2.07813, {True, True}}

Δ is an algebra morphism

In[∘]:=
```
Timing@HL[dm1,2→1 ~ B1 ~ dΔ1→1,2 ≡ (dΔ1→1,3 dΔ2→2,4) ~ B1,2,3,4 ~ (dm3,4→2 dm1,2→1)]
```
Out[∘]= {1.8125, True}

S_2 inverts R, but not S_1:

In[∘]:=
```
Timing@{R1,2 ~ B1 ~ dS1 ≡ R̄1,2, HL[R1,2 ~ B2 ~ dS2 ≡ R̄1,2]}
```
Out[∘]= $\{0.375, \{\frac{1}{4 B_1^3}(4\gamma\epsilon\hbar^2 B_1^2 x_2 y_1 - 2\gamma^2\epsilon^2\hbar^3 B_1^2 x_2 y_1 + 4\gamma\epsilon^2\hbar^3 a_2 B_1^2 x_2 y_1 + 8\gamma^2\epsilon^2\hbar^4 B_1 x_2^2 y_1^2 - 4\gamma\epsilon^2\hbar^4 a_2 B_1 x_2^2 y_1^2 - 3\gamma^2\epsilon^2\hbar^5 x_2^3 y_1^3) == 0, \text{True}\}\}$

S is convolution inverse of id

In[∘]:=
```
Timing[HL[# ≡ 𝔼{1}→{1}[0, 0, 1]] & /@
   {(dΔ1→1,2 ~ B1 ~ dS1) ~ B1,2 ~ dm1,2→1, (dΔ1→1,2 ~ B2 ~ dS2) // dm1,2→1}]
```
Out[∘]= {3.42188, {True, True}}

S is a (co)-algebra anti-morphism

In[∘]:=
```
Timing[HL /@
  Expand /@ {dm1,2→1 ~ B1 ~ dS1 ≡ (dS1 dS2) ~ B1,2 ~ dm2,1→1, dS1 ~ B1 ~ dΔ1→1,2 ≡ dΔ1→2,1 ~ B1,2 ~ (dS1 dS2)}]
```
Out[∘]= {6.75, {True, True}}

Quasi-triangular axiom 1:

In[∘]:=
```
Timing@HL[R1,2 ~ B1 ~ dΔ1->1,3 ≡ (R1,4 R3,2) ~ B2,4 ~ dm2,4→2]
```
Out[∘]= {0.15625, True}

Quasi-triangular axiom 2:

In[∘]:=
```
Timing@HL[((dΔ1->1,2 R3,4) ~ B1,2,3,4 ~ (dm1,3→1 dm2,4→2)) ≡ ((dΔ1→2,1 R3,4) ~ B1,2,3,4 ~ (dm3,1→1 dm4,2→2))]
```
Out[∘]= {1.375, True}

The Drinfel'd element inverse property, $(u_1\,\overline{u}_2)\sim B_{1,2}\sim \mathrm{dm}_{1,2\to 1}\equiv \mathbb{E}[0,0,1]$:

In[∘]:=
```
Timing@HL[((R1,2 ~ B1 ~ dS1 ~ B1,2 ~ dm2,1→i) (R1,2 ~ B2 ~ dS2 ~ B2 ~ dS2 ~ B1,2 ~ dm2,1→j)) ~ Bi,j ~ dmi,j→i ≡
   𝔼{}→{i}[0, 0, 1]]
```
Out[∘]= {1.32813, True}

The ribbon element v satisfies $v^2 = S(u)\,u$. The spinner C=uv^{-1}. It is convenient to compute $z = S(u)\,u^{-1}$ which is something easy.

In[∘]:= $\text{Timing@Block}\left[\{\$k = 2\},\ \left(\left(\left(R_{1,2}\sim B_1\sim dS_1\sim B_{1,2}\sim dm_{2,1\to i}\right)\sim B_i\sim dS_i\right)\ \left(R_{1,2}\sim B_2\sim dS_2\sim B_2\sim dS_2\sim B_{1,2}\sim dm_{2,1\to j}\right)\right)\sim B_{i,j}\sim dm_{i,j\to i}\right]$

Out[∘]= $\left\{1.90625,\ \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \frac{1}{B_i}+\frac{\hbar\, a_i\, \epsilon}{B_i}+\frac{\hbar^2\, a_i^2\, \epsilon^2}{2\, B_i}+O[\epsilon]^3\right]\right\}$

In[∘]:= $\text{Timing@Block}\left[\{\$k = 2\},\ \text{HL /@}\left\{\left(C_i\,\overline{C}_j\right)\sim B_{i,j}\sim dm_{i,j\to i}\equiv \mathbb{E}_{\{\}\to\{i\}}[0,0,1],\ \left(\overline{C}_i\,\overline{C}_j\right)\sim B_{i,j}\sim dm_{i,j\to i}\equiv \left(\left(\left(R_{1,2}\sim B_1\sim dS_1\sim B_{1,2}\sim dm_{2,1\to i}\right)\sim B_i\sim dS_i\right)\ \left(R_{1,2}\sim B_2\sim dS_2\sim B_2\sim dS_2\sim B_{1,2}\sim dm_{2,1\to j}\right)\right)\sim B_{i,j}\sim dm_{i,j\to i}\right\}\right]$

Out[∘]= {2.14063, {True, True}}

Reidemeister 2:

In[∘]:= $\text{Timing}\left[\text{HL}[\# \equiv \mathbb{E}_{\{\}\to\{1,2\}}[0,0,1]]\ \&\ \text{/@}\ \left\{\left(\overline{R}_{1,2}\,R_{3,4}\right)\sim B_{1,2,3,4}\sim\left(dm_{1,3\to1}\,dm_{2,4\to2}\right),\ \left(R_{1,2}\,\overline{R}_{3,4}\right)\sim B_{1,2,3,4}\sim\left(dm_{1,3\to1}\,dm_{2,4\to2}\right)\right\}\right]$

Out[∘]= {1.14063, {True, True}}

Cyclic Reidemeister 2:

In[∘]:= $\text{Timing@HL}\left[\left(R_{1,4}\,\overline{R}_{5,2}\,\overline{C}_3\right)\sim B_{2,4}\sim dm_{2,4\to2}\sim B_{1,3}\sim dm_{1,3\to1}\sim B_{1,5}\sim dm_{1,5\to1}\equiv \overline{C}_1\,\mathbb{E}_{\{\}\to\{2\}}[0,0,1]\right]$

Out[∘]= {0.75, True}

Reidemeister 3:

In[∘]:= $\text{Timing@HL}\left[\left(\left(R_{1,2}\,R_{4,3}\,R_{5,6}\right)\sim B_{1,4}\sim dm_{1,4\to1}\sim B_{2,5}\sim dm_{2,5\to2}\sim B_{3,6}\sim dm_{3,6\to3}\right)\equiv \left(\left(R_{1,6}\,R_{2,3}\,R_{4,5}\right)\sim B_{1,4}\sim dm_{1,4\to1}\sim B_{2,5}\sim dm_{2,5\to2}\sim B_{3,6}\sim dm_{3,6\to3}\right)\right]$

Out[∘]= {1.15625, True}

Relations between the four kinks:

In[∘]:= $\text{Timing}\left[\text{HL /@}\left\{\text{Kink}_i\equiv\left(R_{3,1}\,C_2\right)\sim B_{1,2}\sim dm_{1,2\to1}\sim B_{1,3}\sim dm_{1,3\to i},\ \overline{\text{Kink}}_j\equiv\left(\overline{R}_{3,1}\,\overline{C}_2\right)\sim B_{1,2}\sim dm_{1,2\to1}\sim B_{1,3}\sim dm_{1,3\to j},\ \left(\text{Kink}_i\,\overline{\text{Kink}}_j\right)\sim B_{i,j}\sim dm_{i,j\to1}\equiv\mathbb{E}_{\{\}\to\{1\}}[0,0,1]\right\}\right]$

Out[∘]= {2.1875, {True, True, True}}

The Trefoil

In[∘]:= $\text{Timing@Block}\Big[\{\$k = 1\},\ Z = R_{1,5}\,R_{6,2}\,R_{3,7}\,\overline{C}_4\,\overline{\text{Kink}}_8\,\overline{\text{Kink}}_9\,\overline{\text{Kink}}_{10};\ \text{Do}[Z = Z\sim B_{1,r}\sim dm_{1,r\to1},\ \{r,2,10\}];\ \left\{\text{Simplify /@}\,Z,\ \text{Simplify /@}\left(Z\sim B_1\sim b2t_1\ /.\ T_1\to T\right)\right\}\Big]$

Out[∘]= $\Big\{1.9375,\ \Big\{\mathbb{E}_{\{\}\to\{1\}}\Big[0,\ 0,\ \frac{B_1}{1-B_1+B_1^2}-\left(\hbar\, B_1\left(-a_1\left(-1+B_1-B_1^3+B_1^4\right)+\gamma\left(B_1-2\,B_1^2-2\,B_1^4+2\,\hbar\, x_1\, y_1+B_1^3\left(3+2\,\hbar\, x_1\, y_1\right)\right)\right)\epsilon\right)\Big/\left(1-B_1+B_1^2\right)^3+O[\epsilon]^2\Big],\ \mathbb{E}_{\{\}\to\{1\}}\Big[0,\ 0,\ \frac{T}{1-T+T^2}+\left(T\,\hbar\left(T\left(-1+2\,T-3\,T^2+2\,T^3\right)\gamma+2\left(-1+T-T^3+T^4\right)a_1-2\left(1+T^3\right)\gamma\,\hbar\, x_1\, y_1\right)\epsilon\right)\Big/\left(1-T+T^2\right)^3+O[\epsilon]^2\Big]\Big\}\Big\}$

Program

```
In[∘]:= Define[kR_{i,j} = R_{i,j} // (b2t_i b2t_j) /. t_{i|j} → t,
  k̄R_{i,j} = R̄_{i,j} // (b2t_i b2t_j) /. {t_{i|j} → t, T_{i|j} → T},
  km_{i,j→k} = (t2b_i t2b_j) // dm_{i,j→k} // b2t_k /. {t_k → t, T_k → T, τ_{i|j} → 0},
  kC_i = C_i // b2t_i /. T_i → T,
  k̄C_i = C̄_i // b2t_i /. T_i → T,
  kKink_i = Kink_i // b2t_i /. {t_i → t, T_i → T},
  k̄Kink_i = K̄ink_i // b2t_i /. {t_i → t, T_i → T}]
```

```
In[∘]:= Timing@Block[{$k = 1},
  Z = kR_{1,5} kR_{6,2} kR_{3,7} k̄C_4 k̄Kink_8 k̄Kink_9 k̄Kink_10;
  Do[Z = Z~B_{1,r}~km_{1,r→1}, {r, 2, 10}];
  Simplify /@ Z]
```

Out[∘]= $\left\{1.23438, \mathbb{E}_{\{\}\to\{1\}}\left[0, 0, \frac{T}{1-T+T^2} + \left(T\hbar\left(T\left(-1+2T-3T^2+2T^3\right)\gamma + 2\left(-1+T-T^3+T^4\right)a_1 - 2\left(1+T^3\right)\gamma\hbar x_1 y_1\right)\epsilon\right) \Big/ \left(1-T+T^2\right)^3 + O[\epsilon]^2\right]\right\}$

RVK, rot, Z from 2016-09/OneSmidgen.nb. See also local version in this folder.

Program

Some details of the code below are at http://drorbn.net/bbs/show?shot=Dror-160920-151350.jpg.

Program

```
In[∘]:= RVK::usage =
  "RVK[xs, rots] represents a Rotational Virtual Knot with a list of n Xp/Xm crossings
    xs and a length 2n list of rotation numbers rots. Crossing
    sites are indexed 1 through 2n, and rots⟦k⟧ is the rotation
    between site k-1 and site k. RVK is also a casting operator
    converting to the RVK presentation from other knot presentations.";
```

Program

```
In[∘]:= RVK[pd_PD] := PP_RVK@Module[{n, xs, x, rots, front = {0}, k},
     n = Length@pd; rots = Table[0, {2 n}];
     xs = Cases[pd, x_X :→ { Xp[x⟦4⟧, x⟦1⟧]  PositiveQ@x
                            Xm[x⟦2⟧, x⟦1⟧]     True     ];
     For[k = 0, k < 2 n, ++k, If[k == 0 ∨ FreeQ[front, -k],
       front = Flatten[front /. k → (xs /. {
               Xp[k + 1, l_] | Xm[l_, k + 1] :→ {l, k + 1, 1 - l},
               Xp[l_, k + 1] | Xm[k + 1, l_] :→ (++rots⟦l⟧; {1 - l, k + 1, l})
             })],
        Cases[front, k | -k] /. {k, -k} :→ --rots⟦k + 1⟧;
      ]];
     RVK[xs, rots] ];
RVK[K_] := RVK[PD[K]];
```

```
In[∘]:= xs = Cases[pd, x_X :→ If[PositiveQ@x, Xp[x⟦4⟧, x⟦1⟧], Xm[x⟦2⟧, x⟦1⟧]]];
```

In[•]:=
```
RVK[Knot[10, 100]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[•]=
```
RVK[{Xp[1, 6], Xp[5, 18], Xm[13, 20], Xm[7, 14], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8],
  Xm[19, 12], Xp[17, 2]}, {0, 0, 0, 0, 0, -1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0}]
```

Program

In[•]:=
```
rot[i_, 0] := 𝔼{}→{i}[0, 0, 1];
rot[i_, n_] := Module[{j},
   rot[i, n] = If[n > 0, rot[i, n - 1] kCj, rot[i, n + 1] kC̄j] // kmi,j→i];
```

Program

In[•]:=
```
Z[K_] := Z[RVK@K];
Z[rvk_RVK] := (*Z[rvk] =*)
  PP"Z"@Module[{todo, n, rots, ξ, done, st, cx, ξ1, i, j, k, k1, k2, k3},
    {todo, rots} = List @@ rvk;
    AppendTo[rots, 0];
    n = Length[todo];
    ξ = 𝔼{}→{0}[0, 0, 1];
    done = {0};
    st = Range[0, 2 n + 1];
    While[Echo@todo =!= {},
     {cx} = MaximalBy[todo, Length[done ⋂ {#[[1]], #[[2]], #[[1]] - 1, #[[2]] - 1}] &, 1];
     {i, j} = List @@ cx;
     ξ1 = Switch[Head[cx],
        Xp, (kRi,j kKink̄k) // kmj,k→j,
        Xm, (kR̄i,j kKinkk) // kmj,k→j
       ];
     ξ1 = (rot[k, rots[[i]]] ξ1) // kmk,i→i; rots[[i]] = 0;
     ξ1 = (ξ1 rot[k, rots[[i + 1]]]) // kmi,k→i; rots[[i + 1]] = 0;
     ξ1 = (rot[k, rots[[j]]] ξ1) // kmk,j→j; rots[[j]] = 0;
     ξ1 = (ξ1 rot[k, rots[[j + 1]]]) // kmj,k→j; rots[[j + 1]] = 0;
     ξ *= ξ1;
     If[MemberQ[done, i], ξ = ξ // kmi,i+1→i; st = st /. st[[i + 2]] → st[[i + 1]]];
     If[MemberQ[done, i - 1], ξ = ξ // kmst[[i]],i→st[[i]]; st = st /. st[[i + 1]] → st[[i]]];
     If[MemberQ[done, j], ξ = ξ // kmj,j+1→j; st = st /. st[[j + 2]] → st[[j + 1]]];
     If[MemberQ[done, j - 1], ξ = ξ // kmst[[j]],j→st[[j]]; st = st /. st[[j + 1]] → st[[j]]];
     done = done ⋃ {i - 1, i, j - 1, j};
     todo = DeleteCases[todo, cx]
    ];
    Simplify /@ (ξ /. {x0 → x, y0 → y, a0 → a})
   ]
```

Knot

In[•]:=
```
$k = 1; Timing@Z@Knot[10, 100]
```

Knot

```
{Xp[1, 6], Xp[5, 18], Xm[13, 20], Xm[7, 14],
 Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8], Xm[19, 12], Xp[17, 2]}
```

Knot

```
{Xp[5, 18], Xm[13, 20], Xm[7, 14], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8], Xm[19, 12], Xp[17, 2]}
```

Knot

```
{Xm[13, 20], Xm[7, 14], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8], Xm[19, 12], Xp[17, 2]}
```

Knot

```
{Xm[13, 20], Xm[7, 14], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8], Xm[19, 12]}
```

Knot

```
{Xm[13, 20], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8], Xm[19, 12]}
```

Knot

```
{Xm[13, 20], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[19, 12]}
```

Knot

```
{Xm[13, 20], Xm[3, 10], Xm[11, 4], Xm[19, 12]}
```

Knot

```
{Xm[13, 20], Xm[11, 4], Xm[19, 12]}
```

Knot

```
{Xm[13, 20], Xm[19, 12]}
```

Knot

```
{Xm[13, 20]}
```

Knot

```
{}
```

Knot

Out[∘]=

$$\{36.8125, \mathbb{E}_{\{\}\to\{0\}}[0, 0, T^4 / (1 - 4T + 9T^2 - 12T^3 + 13T^4 - 12T^5 + 9T^6 - 4T^7 + T^8) + (T^4 \hbar (4a(-2 + 14T - 51T^2 + 120T^3 - 203T^4 + 258T^5 - 246T^6 + 152T^7 - 152T^9 + 246T^{10} - 258T^{11} + 203T^{12} - 120T^{13} + 51T^{14} - 14T^{15} + 2T^{16}) + \gamma(-6 + 2T^{16} - 8xy\hbar - 440T^9(-1 + xy\hbar) - 4T^{15}(3 + 2xy\hbar) + 8T^8(-97 + 21xy\hbar) + 8T^7(131 + 21xy\hbar) - 20T^6(57 + 22xy\hbar) + T^{14}(37 + 48xy\hbar) + T(44 + 48xy\hbar) - 8T^{11}(2 + 61xy\hbar) + 8T^5(127 + 68xy\hbar) - 2T^{13}(35 + 78xy\hbar) + 4T^{10}(-39 + 136xy\hbar) - T^2(167 + 156xy\hbar) + T^{12}(79 + 324xy\hbar) + T^3(410 + 324xy\hbar) - T^4(733 + 488xy\hbar)))\epsilon) / (1 - 4T + 9T^2 - 12T^3 + 13T^4 - 12T^5 + 9T^6 - 4T^7 + T^8)^3 + O[\epsilon]^2]\}$$

In[∘]:= **EndProfile**[];

Profile

In[∘]:= **PrintProfile**[]

Profile

Out[∘]=

```
ProfileRoot is root. Profiled time: 80.008
  (     1)   0.264/ 36.797 above Z
  (   157)   0.438/ 34.163 above B
  (    37)   0.112/  8.908 above Boot
  (   147)   0.062/  0.094 above CF
  (     2)   0.046/  0.046 above RVK
CF: called 12152 times, time in 24.968/61.105
  (  1047)   0.798/  3.950 under EEQ
  (  1347)   6.575/ 19.563 under LZip
  (   147)   0.062/  0.094 under ProfileRoot
  (  9611)  17.533/ 37.498 under QZip
  ( 35506)  12.859/ 36.137 above CCF
Together: called 36640 times, time in 17.203/23.496
```

```
  ( 36640)  17.203/ 23.496 under CCF
  ( 36640)   5.511/  6.293 above Exp
CCF: called 36640 times, time in 13.423/36.919
  ( 35506)  12.859/ 36.137 under CF
  (  1134)   0.564/  0.782 under Exp
  ( 36640)  17.203/ 23.496 above Together
Zip: called 2675 times, time in 8.396/39.34
  (   294)   1.015/  6.450 under LZip
  (   294)   0.675/  4.111 under QZip
  (  2087)   6.706/ 28.779 under Zip
  (  2675)   2.165/  2.165 above Collect
  (  2087)   6.706/ 28.779 above Zip
Exp: called 36640 times, time in 5.511/6.293
  ( 36640)   5.511/  6.293 under Together
  (  1134)   0.564/  0.782 above CCF
LZip: called 294 times, time in 5.19/35.53
  (   294)   5.190/ 35.530 under B
  (  1047)   0.377/  4.327 above EEQ
  (  1347)   6.575/ 19.563 above CF
  (   294)   1.015/  6.450 above Zip
Collect: called 2675 times, time in 2.165/2.165
  (  2675)   2.165/  2.165 under Zip
QZip: called 294 times, time in 1.371/42.98
  (   294)   1.371/ 42.980 under B
  (  9611)  17.533/ 37.498 above CF
  (   294)   0.675/  4.111 above Zip
B: called 294 times, time in 0.718/79.228
  (    72)   0.124/ 36.440 under Z
  (    65)   0.156/  8.625 under Boot
  (   157)   0.438/ 34.163 under ProfileRoot
  (   294)   5.190/ 35.530 above LZip
  (   294)   1.371/ 42.980 above QZip
EEQ: called 1047 times, time in 0.377/4.327
  (  1047)   0.377/  4.327 under LZip
  (  1047)   0.798/  3.950 above CF
Boot: called 59 times, time in 0.376/14.079
  (     3)   0.016/  0.093 under Z
  (    19)   0.248/  5.078 under Boot
  (    37)   0.112/  8.908 under ProfileRoot
  (    65)   0.156/  8.625 above B
  (    19)   0.248/  5.078 above Boot
Z: called 1 times, time in 0.264/36.797
  (     1)   0.264/ 36.797 under ProfileRoot
  (    72)   0.124/ 36.440 above B
  (     3)   0.016/  0.093 above Boot
RVK: called 2 times, time in 0.046/0.046
  (     2)   0.046/  0.046 under ProfileRoot
```